// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/extensions/IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.3) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    /**
     * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    /**
     * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
     * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
     */
    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    /**
     * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 oldAllowance = token.allowance(address(this), spender);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance + value));
    }

    /**
     * @dev Decrease the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance - value));
        }
    }

    /**
     * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful. Meant to be used with tokens that require the approval
     * to be set to zero before setting it to a non-zero value, such as USDT.
     */
    function forceApprove(IERC20 token, address spender, uint256 value) internal {
        bytes memory approvalCall = abi.encodeWithSelector(token.approve.selector, spender, value);

        if (!_callOptionalReturnBool(token, approvalCall)) {
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, 0));
            _callOptionalReturn(token, approvalCall);
        }
    }

    /**
     * @dev Use a ERC-2612 signature to set the `owner` approval toward `spender` on `token`.
     * Revert on invalid signature.
     */
    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        require(returndata.length == 0 || abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     *
     * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
     */
    function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
        // and not revert is the subcall reverts.

        (bool success, bytes memory returndata) = address(token).call(data);
        return
            success && (returndata.length == 0 || abi.decode(returndata, (bool))) && Address.isContract(address(token));
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/DoughV2Dsa.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.10;

import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import {IWETH, IDoughV2Index, IAaveV3Pool, IAaveV3DataProvider, CustomError} from "./Interfaces.sol";

contract DoughV2Dsa {
    using SafeERC20 for IERC20;

    /* ========== Layout ========== */
    address public owner;
    address public doughV2Index = address(0);

    /* ========== Constant ========== */
    address private constant _WETH = address(0xC02aaA39b223FE8D0A0e5C4F27eAD9083C756Cc2);
    address private constant _USDC = address(0xA0b86991c6218b36c1d19D4a2e9Eb0cE3606eB48);
    IAaveV3Pool private constant _I_AAVE_V3_POOL = IAaveV3Pool(0x87870Bca3F3fD6335C3F4ce8392D69350B4fA4E2);
    IAaveV3DataProvider private constant _I_AAVE_V3_DATA_PROVIDER = IAaveV3DataProvider(0x7B4EB56E7CD4b454BA8ff71E4518426369a138a3);

    /* ========== CONSTRUCTOR ========== */
    constructor(address _owner, address _doughV2Index) {
        if (_owner == address(0)) revert CustomError("invalid owner");
        if (_doughV2Index == address(0)) revert CustomError("invalid _doughV2Index");
        owner = _owner;
        doughV2Index = _doughV2Index;
    }

    receive() external payable {}

    /* ========== DelegateCall ========== */
    function doughCall(uint256 _connectorId, uint256 _actionId, address _token, uint256 _amount, bool _opt) external payable {
        if (_connectorId == 2) {
            (, , , , , uint256 _healthFactor) = _I_AAVE_V3_POOL.getUserAccountData(address(this));
            (, uint256 _shieldDsaHfTrigger, , ) = IDoughV2Index(doughV2Index).getShieldInfo(address(this));
            if (_healthFactor < _shieldDsaHfTrigger && _actionId == 1) {
                address shieldExecutor = IDoughV2Index(doughV2Index).shieldExecutor();
                if (owner != msg.sender && shieldExecutor != msg.sender) revert CustomError("Ownable: caller is not the owner or Shield executor");
            } else {
                if (owner != msg.sender) revert CustomError("Ownable: caller is not the owner");
            }
        } else {
            if (owner != msg.sender) revert CustomError("Ownable: caller is not the owner");
        }
        address _contract = IDoughV2Index(doughV2Index).getDoughV2Connector(_connectorId);
        if (_contract == address(0)) revert CustomError("doughCall: unregistered connector");
        (bool success, bytes memory data) = _contract.delegatecall(abi.encodeWithSignature("delegateDoughCall(uint256,address,uint256,bool)", _actionId, _token, _amount, _opt));
        if (!success) {
            if (data.length == 0) revert();
            assembly {
                revert(add(32, data), mload(data))
            }
        }
    }

    function executeAction(address tokenIn, uint256 inAmount, address tokenOut, uint256 outAmount, uint256 funcId) external {
        // get connectorV2Flashloan address
        address _connectorV2Flashloan = IDoughV2Index(doughV2Index).getDoughV2Connector(2);
        if (_connectorV2Flashloan == address(0)) revert CustomError("doughV2Dsa: executeAction : unregistered connectorV2Flashloan");
        if (msg.sender != _connectorV2Flashloan) revert CustomError("wrong doughFlashloan");

        //  Loop: funcId = 0 , DeLoop: funcId = 1; Switch: funcId = 2;
        if (funcId > 2) revert CustomError("flashloanReq : invalid-id");

        IERC20(tokenIn).safeTransferFrom(_connectorV2Flashloan, address(this), inAmount);
        IERC20(tokenIn).approve(address(_I_AAVE_V3_POOL), inAmount);
        if (funcId == 0) {
            // Loop
            _I_AAVE_V3_POOL.supply(tokenIn, inAmount, address(this), 0);
            _I_AAVE_V3_POOL.borrow(tokenOut, outAmount, 2, 0, address(this));
            // Check Health Factor
            (, , , , , uint256 _healthFactor) = _I_AAVE_V3_POOL.getUserAccountData(address(this));
            (, uint256 _shieldDsaHfTrigger, , ) = IDoughV2Index(doughV2Index).getShieldInfo(address(this));
            if (_healthFactor <= _shieldDsaHfTrigger) revert CustomError("error: HealthFactor <=  hfTrigger");
        } else if (funcId == 1) {
            // Deloop
            if (tokenIn == _USDC && tokenOut == _USDC) {
                _I_AAVE_V3_POOL.repay(_USDC, inAmount, 2, address(this));
                _I_AAVE_V3_POOL.withdraw(_USDC, outAmount, address(this));
            } else if (tokenIn == _USDC && tokenOut == _WETH) {
                _I_AAVE_V3_POOL.repay(_USDC, inAmount, 2, address(this));
                // get USDC Position
                (uint256 currentATokenBalance, , , , , , , , ) = _I_AAVE_V3_DATA_PROVIDER.getUserReserveData(_USDC, address(this));
                if (currentATokenBalance > 0) {
                    _I_AAVE_V3_POOL.withdraw(_USDC, currentATokenBalance, address(this));
                    IERC20(_USDC).approve(_connectorV2Flashloan, currentATokenBalance);
                }
                _I_AAVE_V3_POOL.withdraw(_WETH, outAmount, address(this));
            }
        } else {
            // Switch
            _I_AAVE_V3_POOL.supply(tokenIn, inAmount, address(this), 0);
            _I_AAVE_V3_POOL.withdraw(tokenOut, outAmount, address(this));
        }
        IERC20(tokenOut).approve(_connectorV2Flashloan, outAmount);
    }
}

// File: contracts/DoughV2Index.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.10;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Address.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "./DoughV2Dsa.sol";

import {CustomError} from "./Interfaces.sol";

contract DoughV2Index is Ownable {
    using Address for address;
    using SafeERC20 for IERC20;

    /* ========== CONSTANTS ========== */
    address private constant _DEAD_ADDRESS = address(0x000000000000000000000000000000000000dEaD);
    IAaveV3Pool private constant _I_AAVE_V3_POOL = IAaveV3Pool(0x87870Bca3F3fD6335C3F4ce8392D69350B4fA4E2);

    /* ========== STATE VARIABLES ========== */
    address public treasury = address(0);
    address public shieldExecutor = address(0);

    uint256 public supplyFee = 0; // 0 %
    uint256 public withdrawFee = 0; // 0 %
    uint256 public borrowFee = 0; // 0 %
    uint256 public repayFee = 0; // 0 %
    uint256 public flashloanFee = 100; // 1 %
    uint256 public shieldFee = 500; // 5 %
    uint256 public shieldSupplyLimit = 15000; // $ 15000

    mapping(address => address) private _doughV2DsaList;
    mapping(uint256 => address) private _connectors;

    struct ShieldInfo {
        uint256 hfTrigger;
        uint256 hfTarget;
        address prevShieldDsa;
        address nextShieldDsa;
    }

    uint256 public shieldDsaCnt = 0;
    address public shieldFirstDsa = address(0);
    address public shieldLastDsa = address(0);
    mapping(address => ShieldInfo) private _shieldDsa;

    /* ========== CONSTRUCTOR ========== */
    constructor(address _treasury) {
        if (_treasury == address(0)) revert CustomError("invalid address");
        treasury = _treasury;
    }

    /* ========== VIEWS ========== */
    function getDoughV2Dsa(address _user) external view returns (address) {
        return _doughV2DsaList[_user];
    }

    function getDoughV2Connector(uint256 _connectorId) external view returns (address) {
        return _connectors[_connectorId];
    }

    function getShieldInfo(address dsa) external view returns (uint256, uint256, address, address) {
        return (_shieldDsa[dsa].hfTarget, _shieldDsa[dsa].hfTrigger, _shieldDsa[dsa].prevShieldDsa, _shieldDsa[dsa].nextShieldDsa);
    }

    /* ========== MUTATIVE FUNCTIONS ========== */
    function withdrawToken(address _tokenAddr, uint256 _amount) external onlyOwner {
        if (_amount == 0) revert CustomError("must be greater than zero");
        uint256 balanceOfToken = IERC20(_tokenAddr).balanceOf(address(this));
        uint256 transAmount = _amount;
        if (_amount > balanceOfToken) {
            transAmount = balanceOfToken;
        }
        IERC20(_tokenAddr).safeTransfer(owner(), transAmount);
    }

    function setTreasury(address _treasury) external onlyOwner {
        if (_treasury == _DEAD_ADDRESS || _treasury == address(0)) revert CustomError("treasury account error");
        treasury = _treasury;
    }

    function setShieldExecutor(address _shieldExecutor) external onlyOwner {
        if (_shieldExecutor == _DEAD_ADDRESS || _shieldExecutor == address(0)) revert CustomError("shieldExecutor account error");
        if (_shieldExecutor == address(0)) revert CustomError("_shieldExecutor is zero_address");
        if (_shieldExecutor == _DEAD_ADDRESS) revert CustomError("_shieldExecutor is zero_address");
        shieldExecutor = _shieldExecutor;
    }

    function setSupplyFee(uint256 _supplyFee) external onlyOwner {
        // _supplyFee <= 5%
        if (_supplyFee > 500) revert CustomError("invaid value");
        supplyFee = _supplyFee;
    }

    function setWithdrawFee(uint256 _withdrawFee) external onlyOwner {
        // _withdrawFee <= 5%
        if (_withdrawFee > 500) revert CustomError("invaid value");
        withdrawFee = _withdrawFee;
    }

    function setBorrowFee(uint256 _borrowFee) external onlyOwner {
        // _borrowFee <= 5%
        if (_borrowFee > 500) revert CustomError("invaid value");
        borrowFee = _borrowFee;
    }

    function setRepayFee(uint256 _repayFee) external onlyOwner {
        // _repayFee <= 5%
        if (_repayFee > 500) revert CustomError("invaid value");
        repayFee = _repayFee;
    }

    function setFlashloanFee(uint256 _flashloanFee) external onlyOwner {
        // _flashloanFee <= 5%
        if (_flashloanFee > 500) revert CustomError("invaid value");
        flashloanFee = _flashloanFee;
    }

    function setShieldFee(uint256 _shieldFee) external onlyOwner {
        // _shieldFee <= 10%
        if (_shieldFee > 1000) revert CustomError("invaid value");
        shieldFee = _shieldFee;
    }

    function setShieldSupplyLimit(uint256 _shieldSupplyLimit) external onlyOwner {
        if (_shieldSupplyLimit == 0) revert CustomError("must be greater than zero.");
        shieldSupplyLimit = _shieldSupplyLimit;
    }

    function setConnectors(uint256 _connectorId, address _connectorsAddr) external onlyOwner {
        if (_connectorsAddr == address(0)) revert CustomError("addConnectors: _connectors address not vaild");
        _connectors[_connectorId] = _connectorsAddr;
    }

    function buildDoughV2Dsa() external returns (address) {
        if (_doughV2DsaList[msg.sender] != address(0)) revert CustomError("created already");
        DoughV2Dsa newDoughV2Dsa = new DoughV2Dsa(msg.sender, address(this));
        _doughV2DsaList[msg.sender] = address(newDoughV2Dsa);
        return address(newDoughV2Dsa);
    }

    function setShield(uint256 _hfTrigger, uint256 _hfTarget) external {
        address dsa = _doughV2DsaList[msg.sender];
        if (dsa == address(0)) revert CustomError("doughV2Index: uncreated dsa");
        if (_hfTrigger <= 1e18) revert CustomError("Shield: wrong trigger value of health factor");
        if (_hfTrigger >= _hfTarget) revert CustomError("Shield: wrong target value of health factor");
        (uint256 totalCollateralBase, , , , , ) = _I_AAVE_V3_POOL.getUserAccountData(dsa);
        if (totalCollateralBase < shieldSupplyLimit * 1e8) revert CustomError("Shield: supply amount < shieldSupplyLimit");
        ShieldInfo storage _refShieldDsa = _shieldDsa[dsa];
        if (shieldDsaCnt == 0) {
            shieldFirstDsa = dsa;
            shieldLastDsa = dsa;
            _refShieldDsa.prevShieldDsa = address(1);
            _refShieldDsa.nextShieldDsa = address(2);
            shieldDsaCnt++;
        } else {
            if (_refShieldDsa.hfTrigger == 0) {
                _shieldDsa[shieldLastDsa].nextShieldDsa = dsa;
                _refShieldDsa.prevShieldDsa = shieldLastDsa;
                _refShieldDsa.nextShieldDsa = address(2);
                shieldLastDsa = dsa;
                shieldDsaCnt++;
            }
        }
        _refShieldDsa.hfTarget = _hfTarget;
        _refShieldDsa.hfTrigger = _hfTrigger;
    }

    function unsetShield() external {
        address dsa = _doughV2DsaList[msg.sender];
        if (dsa == address(0)) revert CustomError("doughV2Index: uncreated dsa");
        if (shieldDsaCnt == 0) revert CustomError("doughV2Index: empty list");

        ShieldInfo storage _refShieldDsa = _shieldDsa[dsa];

        if (_refShieldDsa.hfTrigger <= 1e18) revert CustomError("doughV2Index: already unset");

        if (shieldDsaCnt == 1) {
            shieldFirstDsa = address(0);
            shieldLastDsa = address(0);
        } else {
            if (dsa == shieldLastDsa) {
                _shieldDsa[_refShieldDsa.prevShieldDsa].nextShieldDsa = address(2);
                shieldLastDsa = _refShieldDsa.prevShieldDsa;
            } else if (dsa == shieldFirstDsa) {
                _shieldDsa[_refShieldDsa.nextShieldDsa].prevShieldDsa = address(1);
                shieldFirstDsa = _refShieldDsa.nextShieldDsa;
            } else {
                _shieldDsa[_refShieldDsa.prevShieldDsa].nextShieldDsa = _refShieldDsa.nextShieldDsa;
                _shieldDsa[_refShieldDsa.nextShieldDsa].prevShieldDsa = _refShieldDsa.prevShieldDsa;
            }
            _refShieldDsa.hfTrigger = 0;
            _refShieldDsa.hfTarget = 0;
            _refShieldDsa.prevShieldDsa = address(0);
            _refShieldDsa.nextShieldDsa = address(0);
        }
        shieldDsaCnt--;
    }
}

// File: contracts/Interfaces.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.10;
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

// custom errors
error CustomError(string errorMsg);

// interfaces
interface IWETH is IERC20 {
    function deposit() external payable;

    function withdraw(uint amount) external;
}

interface IDoughV2Index {
    function owner() external view returns (address);

    function treasury() external view returns (address);

    function shieldExecutor() external view returns (address);

    function supplyFee() external view returns (uint256);

    function withdrawFee() external view returns (uint256);

    function borrowFee() external view returns (uint256);

    function repayFee() external view returns (uint256);

    function flashloanFee() external view returns (uint256);

    function shieldFee() external view returns (uint256);

    function getShieldInfo(address dsa) external view returns (uint256, uint256, address, address);

    function getDoughV2Connector(uint256 _connectorId) external view returns (address);
}

interface IDoughV2Dsa {
    function doughV2Index() external view returns (address);

    function executeAction(address tokenIn, uint256 inAmount, address tokenOut, uint256 outAmount, uint256 funcId) external;
}

interface IAaveV3DataProvider {
    function getUserReserveData(
        address asset,
        address user
    ) external view returns (uint256 currentATokenBalance, uint256 currentStableDebt, uint256 currentVariableDebt, uint256 principalStableDebt, uint256 scaledVariableDebt, uint256 stableBorrowRate, uint256 liquidityRate, uint40 stableRateLastUpdated, bool usageAsCollateralEnabled);
}

interface IAaveV3Pool {
    function FLASHLOAN_PREMIUM_TOTAL() external view returns (uint128);

    function getUserAccountData(address user) external view returns (uint256 totalCollateralBase, uint256 totalDebtBase, uint256 availableBorrowsBase, uint256 currentLiquidationThreshold, uint256 ltv, uint256 healthFactor);

    function supply(address asset, uint256 amount, address onBehalfOf, uint16 referralCode) external;

    function withdraw(address asset, uint256 amount, address to) external;

    function borrow(address asset, uint256 amount, uint256 interestRateMode, uint16 referralCode, address onBehalfOf) external;

    function repay(address asset, uint256 amount, uint256 interestRateMode, address onBehalfOf) external returns (uint256);
}

interface IConnectorV2Flashloan {
    function flashloanReq(address _loanToken, uint256 _loanAmount, uint256 _funcId, bool _isShield) external;
}

interface IUniswapV2Router {
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);

    function swapExactTokensForTokens(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline) external returns (uint[] memory amounts);
}